PS/8 NEWSLETTER

Submissions should be sent to:

J. C. Alderman
Georgia Tech Research Reactor
900 Atlantic Drive, N.W.
Atlanta, Georgia 30318

RECENT CONTRIBUTIONS:

MAGTAPE HANDLER FOR PS/8 (7 TRACK, TC-58)
by J. Alderman

Georgia Tech Research Reactor
900 Atlantic Drive, N.W.
Atlanta, Georgia 30318

The program for the Handler is given along with a flow chart. The Magtape must be formatted and two FOCAL formatter programs are also given.

```
/MTA HANDLER FOR PS/8 MONITOR
/DIRECTORY VERSION, IN ONE PAGE!
/
/EACH "BLOCK" ON TAPE CONSISTS OF 3 RECORDS.
/       1. A SINGLE WORD RECORD CONTAINING THE
/          BLOCK NUMBER.
/
/       2. 256 or 128 WORDS OF DATA.
/
/       3. AN END-OF-FILE RECORD (EOF=17)
/
/
/THERE IS A TAPE FORMATTER PROGRAM (WRITTEN IN
/FOCAL !!) TO
/FORMAT A TAPE WITH THE REQUIRED BLOCK STRUCTURE.
/
/NOTE: BECAUSE OF THE WAY THE FORMATTED TAPE IS
/GENERATED,
/EACH "BLOCK" (GROUP OF 3 RECORDS AS DESCRIBED
/ABOVE)
/IS ONLY 3.75 INCHES LONG.  THUS A 200' REEL OF
/TAPE
/WILL CONTAIN 640 BLOCKS.
/
/
/ASSEMBLY PARAMETERS
/
EOF=    0100
FORWAR= 0060
REVERS= 0070
REWIND= 0010
WRITE=  0040
TEST=   3766
READ=   0020
/
/MAGTAPE IOT'S
/
MTSF=6701
MTCM=6714
MTCR=6711
MTTR=6721
MTAF=6712
MTRC=6724
MTLC=6716
MTRS=6706
MTGO=6722
MCAF=6732
*7200
/
/ENTRY
/
MTA,    0           /SINGLE ENTRY ONLY!
        RDF
        TAD CDFCIF
        DCA ERROR
        RIF         /WHAT IS THE FIELD WE'RE IN?
        TAD KCDF
        DCA MFIELD+1
        TAD I MTA
        DCA MTFUNC
        ISZ MTA
        TAD I MTA
        DCA MTADDR
        ISZ MTA
        TAD I MTA
        DCA BLOCK
        ISZ MTA
        JMS MFIELD
        MCAF        /POWER CLEAR TAPE CONTROL.

        TAD MTFUNC
        AND P3700   /SAVE PAGECOUNT.
        CLL RTL
        RTL
        RTL         /BLOCKS IN ACC, EVEN/ODD IN
                    /LINK.
        SZL         /SKIP IF EVEN # PAGES
        IAC
        CIA
        DCA BLCNTR  /A WELL-ROUNDED COUNT!
        SZL         /SKIP IF EVEN # PAGES.
        TAD BLCNTR
        DCA HALFBLOCK  /WATCH THIS, IT IS SUPER-
                       /TRICKY!
        JMP NEXTBL
SEARCH, SMA         /FIND THE BLOCK.
        JMP NEXTBLOCK-1
        TAD (-1
        DCA BCNTR
        STA CLL RTL  /-3
        TAD BLOCK
        SNL CLA      /SKIP IF BLOCK< 3
        JMP SPACE
        DCA BCNTR
        TAD (REWIND-READ
        JMP NEXTBLOCK
SPACE,  TAD (REVERSE-READ
        JMS MOVE
P200,    200
PREAD,   READ
MTWC,    7752
         EOF
        ISZ BCNTR   /SKIP IF FINISHED SPACING
        JMP SPACE
        CLA CLL
NEXTBL, JMS MOVE
         1
MTCA,    7753
BCNTR,   0          /COUNTS BLOCK-SPACING, ALSO 0.
         TEST
        JMP NEXTBLOCK  /BLOCK RECORD INCORRECT.
MTLOOP, CDF
        TAD I MTCA
        JMS MFIELD
        CIA CLL
        TAD BLOCK
        SZA         /SKIP IF BLOCK FOUND.
        JMP SEARCH
        ISZ HALFBLOCK  /SKIP IF LASTBLOCK AND #
                       /PAGES IS OD
        TAD P200    /FULL BLOCK.
        TAD P200
        DCA MTADDR-1
```

```
        TAD MTFUNC  /GET FUNCTIONWORD AGAIN.
        SPA CLA     /SKIP IF READ IS FUNCTION
        TAD PREAD
        JMS MOVE
MTCOMM,  400
MTADDR,  0          /ADDRESS OF DATA.
MTFUNC,  0          /FUNCTIONWORD.
         TEST
        JMP ERROR   /TRANSFER BOMBED.
        TAD MTADDR
        TAD MTADDR-1  /UPDATE DATA ADDRESS.
        DCA MTADDR
        ISZ BLOCK   /STEP TO NEXT BLOCK IN FILE.
        ISZ BLCNTR  /SKIP WHEN TRANSFER COMPLETE.
        JMP NEXTBLOCK
        ISZ MTA     /INDICATE NO ERROR.
ERROR,  CDF CIF
        JMP I MTA

/MOVE.  CALLED BY:
/       JMS MOVE    /ACC CONTAINS FUNCTION IN BITS
                    /6-8.
/        WORDCOUNT       /NEGATED INTERNALLY.
/        CURRENT ADDRESS /-1 SUBTRACTED INTERNALLY
/        FIELD
/        STATUS BIT PATTERN
/       RETURN IF MASKED STATUS #0.
/       RETURN IF MASKED STATUS =0.
MOVE,   0
        TAD UPD
        MTLC
        MTRS
        TAD M400
        SNA CLA     /SKIP IF TRANSPORT NOT
                    /SETTLING.
        JMP .-3
        TAD I MOVE  /WORDCOUNT
        CIA
KCDF,   CDF
        DCA I MTWC
        JMS MFIELD
        STA CLL
        TAD I MOVE  /ADDRESS
        CDF
        DCA I MTCA
        JMS MFIELD
        TAD I MOVE  /FIELD
        ISZ MOVE
        MTGO
        MTTR
        JMP .-1
        MTRS
        AND I MOVE  /MASK
        ISZ MOVE
        SNA CLA     /SKIP IF STATUS!MASK#0
        ISZ MOVE
        JMP I MOVE

MFIELD, 0
        CDF
        ISZ MOVE    /SAVE 1 LOCATION IN MOVE!
        JMP I MFIELD
P3700,   3700       /CONSTANT.
BLOCK,   0          /CONTAINS DESIRED BLOCK.
CDFCIF, CDF CIF     /CONSTANT.
UPD,    421         /UNIT 0, ODD PARITY, 556
                    /BPIREAD COMMAND.
M400,   -400        /CONSTANT.
HALFBL, 0           /DECIDES ABOUT FINAL BLOCK
                    /LENGTH.
BLCNTR, 0           /COUNTS BLOCKS.
```

MAGTAPE FORMATTER FOR FOCAL UIOF

```
C-UIOF 7AUG 69

01.01 C GETS PARAMETERS FOR TAPE
01.10 S T !"LENGTH OF TAPE IN FEET
01.20 A LE;S LE=(LE*12/3.75)-1;I (LE-4095)1.3;S
       LE=4095
01.30 D 11;S I=0;D 15;D 14;D 15;S I=1;D 14
01.40 S Z=FDXS(4096,4095)+FDXS(4097,7)+FDXS(4098,0)
       +FDXS(4099,0)
01.50 S Z=FDXS(4100,0)+FDXS(4101,0)
01.60 S Z=FDXS(4102,4103-LE)+FDXS(4074,3840)+
       FDXS(4075,4095)
01.70 D 14.2;D 13;D 15;F I=2,LE;D 14;D 15
01.80 D 11;T %4,LE," BLOCKS WRITTEN",!;Q

11.01 C REWINDS THE TAPE.
11.10 S Z=FADC(3546)+FADC(3534,265)+FADC(3538)
11.20 D 13

13.01 IF (-FADC(3521))13.01,13.01;RR

14.01 C WRITES A ONE-WORD BLOCK NUMBER RECORD.
14.10 S Z=FDXS(4096,I)+FDXS(4074,4095)+FDXS
       (4075,4095)
14.20 S Z-FADC(3534,289)+FADC(3538,8)
14.30 D 13

15.01 C WRITES AN END-OF-FILE.
15.10 S Z=FADC(3534,297)+FADC(3538)
15.20 D 13
*
```

MAGTAPE FORMATTER FOR FOCAL S

```
C-FOCAL S 1/71

01.01 C GETS PARAMETERS FOR TAPE
01.10 S BASE=4096+2048;L L,2048
01.18 T !"LENGTH OF TAPE IN FEET
01.20 A LE;S LE=(LE*2)-1;I (LE-4095)1.3;S LE-4095
01.30 D 11;S I=0;D 15;D 14;D 15;S I=1;D 14
01.40 S Z=FX(1,BASE,-1)+FX(1,BA+1,7)+FX(1,BA+2,0)+
       FX(1,BA+3,0)
01.50 S Z=FX(1,BA+4,-1)+FX(1,BA+5,0)+FX(1,BA+6,
       7-LE)
01.60 S Z=FX(1,4074,3840)+FX(1,4075,BA-4096-1)
01.70 D 14.2;D 13;D 15;F I=2,LE;D 14;D 15
01.80 D 11;T %4,LE," BLOCKS WRITTEN",!;Q

11.01 C REWINDS THE TAPE.
11.05 I (0*FX(2,3526)+FX(3,FX(2,3526),FX(10,1000))
       ),11.1;R
11.10 S Z=FX(2,3534,265)+FX(2,3538)
11.20 D 13

13.10 I (FX(3,FX(2,3526),1)+FX(3,FX(2,3526),1)
       -2)13.1;R

14.01 C WRITES A ONE-WORD BLOCK NUMBER RECORD.
14.10 I (FX(1,BA+1)+FX(1,4074,-1)+FX(1,4075,BA
       -4096-1))
14.20 I (FX(2,3534,289)+FX(2,3538,8))
14.30 D 13

15.01 C WRITES AN END-OF-FILE.
15.10 I (FX(2,3534,297)+FX(2,3538))
15.20 D 13
*
```

FLOW CHART: MAGTAPE HANDLER FOR PS/8



SERIAL HIGH-SPEED DATA-LINK FOR DP12 OR PT08

by J. Alderman
Georgia Tech Research Reactor
900 Atlantic Drive, N.W.
Atlanta, Georgia 30318

This Handler can be used in place of the CDR Handler. The listing is given below.

```
/SERIAL HIGH-SPEED DATA LINK
/EACH END MUST ECHO AS RECEIVING.

        IFNDEF INDEV < INDEV=6400 >
        IFNDEF OUTDEV < OUTDEV=6410 >
        INSF=INDEV+1;INCC=INDEV+2;INRS=INDEV+4;
          INRB=INDEV+6
        OUTSF=OUTDEV+1;OUTCF=OUTDEV+2;
          OUTLS=OUTDEV+6
        FIELD 0
        *5600           /REPLACING CDR: HANDLER.
PTP,    0
        CLA CLL
        TAD I PTP       /FUNCTION
        SMA CLA         /SKIP IF SENDING.
        TAD (JMP PTRLP
        DCA PTPCNT      /EITHER A JMP, OR 0.
        JMS PSETUP      /DO COMMON CRAP
PTFLP,  KSF
        JMP PTPCNT      /KEYBOARD FLAG OFF
        KRS
        AND (177
        TAD PTPM3
        SZA CLA  /IS THERE A ↑C IN THE TTY BUFFER?
        JMP PTPCNT      /NO
PTPCIF, CDF CIF 0
        JMP I PT7600
PTPCNT, JMP PTRLP       /0 IF SENDING.
        TAD I PTPCA
        JMS PTPPCH      /FIRST CHAR IN LOW ORDER 8
                        /BITS OF WORD
        DCA PTR
        ISZ PTPCA
PT7700, 7700            /SMA CLA, BUT JUST CLEARS ACC.
        TAD I PTPCA
        JMS PTPPCH      /SECOND CHAR IN LOW ORDER 8
                        /BITS OF WORD
        CLL RTR
        RTR
        TAD PTR
        RTR
        RTR         /THIRD CHARACTER NOW IN AC
        JMS PTPPCH
PTPEND, ISZ PTPCA
PT70,   70          /JUST IN CASE WERE PUNCHING
                    /PG 7600 KEEP THIS
        JMP PTPISZ
PSETUP, 0
        RDF         /GET FIELD OF CALLING PROGRAM
        TAD PTPCIF
        DCA PTPXIT      /SET UP RETURN SEQUENCE
        TAD I PTP
        AND PT70
        TAD PCDF
        DCA PTPCDF
        TAD I PTP       /GET FUNCTION WORD
        ISZ PTP
        AND PT7700
        CMA             /SET UP -(WORD COUNT)/2-1
        DCA PTPWC
        TAD I PTP       /SET UP STARTING ADDRESS
        ISZ PTP
        DCA PTPCA
PTPCDF, 0               /SET DATA FIELD TO ACCESS
                        /BUFFER
        PTPEOF=PTPCDF
PT7600, 7600
PTPISZ. ISZ PTPWC
        JMP I PSETUP /LOOP FOR BUFFER SIZE
                     /(128 WORDS)
PTPRTN, TAD PTPEOF
        SZA CLA      /DID WE RUN OUT OF TAPE?
        ISZ PTP
        ISZ PTP
PTPXIT, HLT      /EXIT CDF GOES HERE
        JMP I PTP
PTPPCH, 0
        OUTLS
        DCA PTMP
        INCC         /CLEAR OLD ECHO CHARACTER.
        TAD PTMP
        AND (377
        CIA
        DCA PTMP1
        INSF
        JMP .-1
        INRB
        TAD PTMP1
        SZA CLA      /SKIP IF SAME CHARACTER
                     /ECHOED.
        JMP PTPERR
        TAD PTMP
        AND PT7400
        JMP I PTPPCH
PTPCA,  0
PTR,    0
PTRLP,  JMS PTRGCH   /READ FIRST CHARACTER OF 3
        DCA I PTPCA
        JMS PTRGCH
        DCA PTPPCH
        JMS PTRGCH
        RTL
        RTL
        DCA PTR
        TAD PTR
        AND PT7400
        TAD I PTPCA
        DCA I PTPCA  /HIGH ORDER 4 BITS INTO WORD 1
        TAD PTR      /GET THE CHAR FROM THE PTR
                     /BUFFER
        RTL
        RTL
        AND PT7400
        TAD PTPPCH
        ISZ PTPCA
PT7400, 7400
        DCA I PTPCA  /LOW ORDER 4 BITS INTO WORD 2
        JMP PTPEND
PTRGCH, 0
        INSF
        JMP PTRTST   /CHECK EOF.
        INRB
        OUTLS        /PRODUCE ECHO.
        JMP I PTRGCH /RETURN WITH CHARACTER
PTRIST, KRS
        TAD (-232    /-↑Z
        SZA CLA      /SKIP IF EOF.
        JMP PTRGCH+1 /LOOK AGAIN.
        KRS          /PICKUP ↑Z.
        JMP I PTRGCH
PTPERR, CLA CLL CML RAR /SIGNAL A "PERMANENT I/O
                        /ERROR" ON TE
        JMP PTPXIT-1
PCDF,   CDF 0
PTPWC,  0
```

```
PTPM3,  -3
PTMP,   0
PTMP1,  0


        *7576
        0220            /READ & WRITE, NON-DIRECTORY
                        /DEVICE.
        FIELD 1
        *1004           /SPECIFIC TO PS-8 6-NOV-70
                        /RELEASE!
        3↑200+PTP-5600  /REPLACES CDR IN BLOCK 21
        *602
        7224      /LINK INSTEAD OF CDR
        $
```

LOW SPEED PTR: AND PTP: USING THE TELETYPE

by J. Alderman
Georgia Tech Research Reactor
900 Atlantic Drive, N.W.
Atlanta, Georgia 30318

This Handler replaces the PTR/PTP Handler in CONFIG for those people without the high-speed reader/punch.

```
/PTR: & PTP: USING TELETYPE
/FOR PEOPLE WHO HAVE NO HIGH SPEED PAPER TAPE.

        FIELD 0
        *6400
PTP,    0
        CLA CLL CML  /SET LINK ON TO INDICATE
                     /PUNCH
        JMS PSETUP   /DO COMMON CRAP
PTPLP,  KSF
        JMP PTPCNT   /KEYBOARD FLAG OFF
        KRS
        AND PTP177
        TAD PTPM3
        SZA CLA  /IS THERE A↑C IN THE TTY BUFFER?
        JMP PTPCNT   /NO
PTPCIF, CDF CIF 0
        JMP I PT7600
PTPCNT, TAD I PTPCA
        JMS PTPPCH   /FIRST CHAR IN LOW ORDER 8
                     /BITS OF WORD
        DCA PTR
        ISZ PTPCA
PT7700, 7700
        TAD I PTPCA
        JMS PTPPCH   /SECOND CHAR IN LOW ORDER 8
                     /BITS OF WORD
        CLL RTR
        RTR
        TAD PTR
        RTR
        RTR      /THIRD CHARACTER NOW IN AC
        JMS PTPPCH
PTPEND, ISZ PTPCA
PT70,   70       /JUST IN CASE WERE PUNCHING PG
                 /7600 KEEP THIS
        JMP PTPISZ
PSETUP, 0
        RDF      /GET FIELD OF CALLING PROGRAM
        TAD PTPCIF
        DCA PTPXIT   /SET UP RETURN SEQUENCE
        TAD I PTP
        AND PT70
        TAD PCDF
        DCA PTPCDF
        RAR          /GET LINK(1=PTP,0=PTR)
        TAD I PTP    /GET FUNCTION WORD
        ISZ PTP
        SPA          /CHECK CORRECT MODE
        JMP PTPERR   /SIGNAL "UNRECOVERABLE DEVICE
                     /ERROR"
        AND PT7700
        CMA          /SET UP -(WORD COUNT)/2-1
        DCA PTPWC
        TAD I PTP    /SET UP STARTING ADDRESS
        ISZ PTP
        DCA PTPCA
        TAD I PTP
PTPCDF, 0            /SET DATA FIELD TO ACCESS
                     /BUFFER
        PTPEOF=PTPCDF
        SNA CLA
        SZL
        JMP PTPISZ
        TAD PTP336   /INPUT INITIALIZATION - TYPE
                     /"↑T"
        TLS
        KSF
        JMP .-1
PT7600, 7600
PTPISZ, ISZ PTPWC
        JMP I PSETUP /LOOP FOR BUFFER SIZE (128
                     /WORDS)
PTPRTN, TAD PTPEOF
        SZA CLA      /DID WE RUN OUT OF TAPE?
        ISZ PTP
        ISZ PTP
PTPXIT, HLT      /EXIT CDF GOES HERE
        JMP I PTP
PTPPCH, 0
        TLS
        TSF          /NOTICE THE GLORIOUS LACK OF
                     /OVERLAP
        JMP .-1
        AND PT7400
        JMP I PTPPCH
PTPCA,  0
PTR,    0            /CORRECT ENTRY IN MAIN
                     /ASSEMBLY
        CLA CLL
        TAD PTR
        DCA PTP
        JMS PSETUP   /SET UP ADDRESS, COUNT, FIELDS
PTRLP,  JMS PTRGCH   /READ FIRST CHARACTER OF 3
        DCA I PTPCA
        JMS PTRGCH
        DCA PTPPCH
        JMS PTRGCH
        RTL
        RTL
        DCA PTR
        TAD PTR
        AND PT7400
        TAD I PTPCA
        DCA I PTPCA  /HIGH ORDER 4 BITS INTO WORD 1
        TAD PTR      /GET THE CHAR FROM THE PTR
                     /BUFFER
        RTL
        RTL
        AND PT7400
        TAD PTPPCH
        ISZ PTPCA
PT7400, 7400
        DCA I PTPCA  /LOW ORDER 4 BITS INTO WORD 2
        JMP PTPEND
PTRGCH, 0
        TAD PTPEOF
        SNA CLA
        JMP I PTRGCH
        CLA CLL CMA RTL      /-3
```

```
        DCA PTPEOF
        DCA PTR
PTTIME, ISZ PTR         /TIMEOUT LOOP FOR LOW SPEED
                        /READER
        JMP PTP232
        ISZ PTPEOF      /TIMES OUT IN 132 MS(PDP 8/E)
                        /OR 205 MS
        JMP PTP232
        TAD PTP232
        JMP I PTRGCH /OVERFLOWED - PTPEOF IS NOW 0,
                        /RETURN
PTP232, 232             /WASTE SOME TIME
PTP177, 177
        KSF
        JMP PTTIME      /READER NOT READY - CONTINUE
                        /TIMEOUT
        KRB
        JMP I PTRGCH /RETURN WITH CHARACTER
PTPERR, CLA CLL CML RAR  /SIGNAL A "PERMANENT I/O
                         /ERROR" ON TE
        JMP PTPXIT-2
PCDF,   CDF 0
PTP336, 336
PTPNC,  0
PTPM3,  -3

        FIELD 1
        *1003           /SPECIFIC TO PS-8 6-NOV-70
                        /RELEASE!
        1↑200+PTR-6400
        $
```

PDP-12 DECTAPE CONVERT PROGRAM

by J. Alderman
Georgia Tech Research Reactor
900 Atlantic Drive, N.W.
Atlanta, Georgia 30318

For PDP-12 users a listing of the changes needed to convert DECtape by changing PRTC12-F to make it JATC12-F, (PRTC12-F doesn't do this for PS/8 or IM LINCtapes) is given below.

```
0365  7770  M7,       -7
0366  0400  K400,     400
0367  2653  MODOP4,   ADD P8FMT2  /PDP-12 to PDP-8
0370  4630            STC WRILNG  /READ 201 WORDS
0371  2624            ADD P8ENUM  /WRITE 201 WORD
0372  4632            STC WRINUM
0373  2652            ADD P8FMT1  /201 WORDS/BLOCK
0374  1040            STA
0375  2747            ADDINC!2000
0376  1020            LDA I
0377  7761            -16
0400  1040            STA
0401  2066            RDLTAN!2000
0402  1040            STA
0403  2037            WDLTAN!2000
0404  0011            CLR
0405  0017            COM
0406  1040            STA
0407  2070            RDEKEY!2000  /7777=READ FROM
                                   /LINCT
              APE
0410  1020            LDA I
0411  7575            -202
0412  1040            STA
0413  2041            WRIKEY!2000
0414  6616            JMP GO
              /
              /
              /2/5/71 FOR 201 BLOCK LINCTAPE
              /MODOP4 CHANGED.
```

Use of JATC12-F with PS/8 LTA:

```
LO JATC12-F,

2000  BLOCKS
B     PS/8 LTA
0     FROM UNIT 0
0     1st BLOCK =0
A     PDP-8 DTA
1     UNIT 1
0     1st BLOCK =0
0     NO CHECK
702   BLOCKS
B
0
2000
A      SAME
1
2000
0
2000  BLOCKS
A     PDP-8 PS/8 DTA
1     UNIT 1
0     1st BLOCK =0
B     PS/8 LTA
0     UNIT 0
0     1st BLOCK =0
1     CHECK PARITY
702
A
1
2000
B      SAME
0
2000
1
```

With this issue we are circulating the first printing of the PS/8 Newsletter. In the future these newsletters will be sent to a special mailing list of PS/8 users only. This PS/8 list is now in the process of being compiled. Anyone interested in being added to this list should contact the DECUS office.

DECUSCOPE HAS BEEN PUBLISHED SINCE APRIL 1962 AND IS THE OFFICIAL NEWSLETTER FOR DIGITAL EQUIPMENT COMPUTER USERS SOCIETY.

IT IS PUBLISHED PERIODICALLY AT THE DECUS OFFICE, DIGITAL EQUIPMENT CORPORATION, MAYNARD, MASSACHUSETTS 01754.

TELEPHONE: Area Code 617, 897-5111, Ext. 2414

EDITOR: ANGELA J. COSSETTE, DECUS

CIRCULATION: 10,500 PER ISSUE